
Software concepts





Concepts

• Apps are made of recurring concepts 

• Each concept is a self-contained unit of functionality with a clear purpose


• Concepts work together to provide the app overall functionality


• But can be understood independently of one another

Concept design

• Identify a clear purpose


• Choose the appropriate state and actions to fulfill that purpose 

• The focus is on ensuring correctness and reusability



App design

• Identify the core concepts


• Compose them, maybe providing new functionality


• The focus is on exploration



Modeling Concepts



Trash



Trash



Trash



Trash



Trash modeling a la UML

Trash modeling a la Jackson
concept trash [Item]
purpose
  to allow undoing of deletions
state
  accessible, trashed : set Item
actions
  create (x : Item)
    when x not in accessible or trashed
    add x to accessible
  delete (x : Item)
    when x in accessible but not trashed
    move x from accessible to trashed
  restore (x : Item)
    when x in trashed
    move x from trashed to accessible
  empty ()
    when some item in trashed
    remove every item from trashed
operational principle
  after delete(x), can restore(x) and then x in accessible
  after delete(x), can empty() and then x not in accessible or trashed



Concept modeling a la Jackson
• Name


- Optionally parametrized by types that can be specialized when composing


• Purpose


- A clear reason why you might want it


• State + Actions


- A description of the concept behavior using a transition system


• Operational principle


- Properties that show how the purpose is fulfilled by the actions



Transition systems



Transition systems
• A popular model to describe the behavior of a system


• A model is often a synonym for a transition system


• There are many variants and related formalisms


- Labeled transition systems


- Kripke structures


- Finite state machines


- Hybrid and timed automata


- ...



States, transitions, and traces
• States


- A state is a possible valuation to the structures of the system


- Initial states describe how the system starts


• Transitions


- A transition is a possible evolution between states


- Transitions originate from actions of the system or the environment


• Traces


- A trace is a sequence of states, describing a possible execution


- A valid trace in a transition system is a path starting in an initial state



Trash transition system



Declarative modeling

• It is possible to describe a transition system by specifying instead which 
are its valid traces


• This requires specifying a property whose validity is established in a trace 
and not just in a single state


• The specification of properties about traces requires some sort of 
temporal logic



Valid trash traces



Invalid trash traces



Specifying transition systems



Mutability

• In Alloy 6 mutable signatures and fields can be declared with keyword var


- Static field inside mutable signature yields a warning


- Static signature extending or subset of a mutable one also yields a 
warning



Trash states

sig Item {}
var sig Accessible in Item {}
var sig Trashed in Item {}



Instances
• When mutable structures are declared, instances are infinite traces 

• Analysis commands only return traces that can be represented finitely


- Traces that loop back at some point


• Static signatures and fields are known as the configuration and have the 
same value in all states


• If there are mutable top-level signatures, univ (and iden) are also 
mutable



Temporal logic
• Alloy 6 also supports linear temporal logic


• Temporal logic adds temporal operators to relational logic 

• They allow us to “quantify” the validity of a formula over the different states of 
a trace


• A formula without temporal operators is only required to hold in the initial 
states


• Alloy 6 has both future and past temporal operators


• It also has the prime operator that denotes the value of a term in the next state



Always, eventually, and prime

always            is true in all future states
eventually        is true in some future state
'                The value of  in the next state

ϕ ϕ
ϕ ϕ

R R



Actions
• A set of transitions can be specified declaratively with an action


- A formula without temporal operators, but including primed and unprimed variables


- A condition without primed variables is a guard that specifies when is the action enabled


- A condition with a primed variable is an effect that specifies what are the possible 
values for that variable after the action occurs


- If a variable does not change, a frame condition should be included stating that the next 
value of the variable is the same


• By combining actions with the always temporal operator we can specify a system behavior


• Actions were first introduced by Leslie Lamport in the Temporal Logic of Actions



Empty trash



Empty trash

pred empty {
  // guard
  some Trashed
  // effect
  no Trashed'
  // frame condition
  Accessible' = Accessible
}



Create item



Create item

pred create [i : Item] {
  // guard
  i not in Accessible + Trashed
  // effect
  Accessible' = Accessible + i
  // frame condition
  Trashed' = Trashed 
}



Delete item



Delete item

pred delete [i : Item] {
  // guard
  i in Accessible
  // effects
  Accessible' = Accessible - i
  Trashed' = Trashed + i
}



Restore item



Restore item

pred restore [i : Item] {
  // guard
  i in Trashed
  // effects
  Accessible' = Accessible + i
  Trashed' = Trashed - i
}



Trash behavior



Trash behavior
fact Behavior {
  // initial state
  no Accessible
  no Trashed
  // possible transitions
  always {
    (some i : Item | create[i] or delete[i] or restore[i]) 
    or
    empty
  }
}



Validation



Run commands

• As usual, run commands can be used to validate the model


• The scope of a mutable signature defines the maximum number of 
different atoms in the full trace, not a maximum per state



Trace visualization
• The visualizer depicts two consecutive states of the trace side-by-side


- By default mutable structures are depicted with dashed lines


• A representation of the infinite trace is shown above


- Different states have different numbers


- The loop back is explicitly depicted


- Clicking on a state focus on that (and the succeeding) state


- It is possible to move forwards and backwards in the trace with → and ←



Trace visualization

Simulation

• It possible to perform “simulation” with the New instance buttons


- New config, returns a trace with a different configuration (a different 
value to the immutable structures)


- New trace, returns any different trace with the same configuration


- New init, returns a trace with the same config, but a different initial state


- New fork, returns a trace with the same prefix, but a different next state



Simulation

DEMO



Specifying scenarios

• A formula can be given in a run command to look for specific scenarios 


• Keyword expect can be used to distinguish positive and negative 
scenarios



Semi-colon

 ;              is valid after ϕ ψ ψ ϕ



Some trash scenarios
run Scenario1 {
  some i : Item {
    create[i]; delete[i]; restore[i]; delete[i]; empty
  }
} expect 1

run Scenario2 {
  some disj i,j : Item {
    create[i]; delete[j]
  }
} expect 0

run Scenario3 {
  some i : Item {
  create[i]; delete[i]; empty
}

} for 1 Item expect 1



Stuttering



A clock specification

pred clock_spec {
  h = 12 and m = 0
  always {
    m’=(m+1)%60 and
    m=59 implies h’=(h%12)+1 else h’=h
 }
}



Ceci n'est pas une montre?!

check clock_spec



A clock specification
pred clock_spec {
  h = 12 and m = 0
  always {
    m’=(m+1)%60 and
    m=59 implies h’=(h%12)+1 else h’=h
    or
    m’=m and h’=h
 }
}



Another clock

check clock_spec



Stuttering

• Stuttering can represent events by the environment or by other 
components of the system (not yet modeled)


• Stuttering enables refinement


- adding detail or new components to a system


- namely, it enables concepts to be composed to build apps


• In terminating systems, stuttering enables traces to be infinite



Trash stuttering

pred stutter {
Accessible' = Accessible
Trashed' = Trashed

}



Trash behavior
fact Behavior {
  // initial state
  no Accessible
  no Trashed
  // possible transitions
  always {
    (some i : Item | create[i] or delete[i] or restore[i]) 
    or
    empty
    or
    stutter
  }
}



Verification



Model checking
• Model checking is the process of automatically verifying if a temporal logic 

specification holds in a finite transition system model of a system


- If the specification is false a counter-example is returned


- A finite transition system may have infinite non-looping traces


- But every invalid specification can be falsified with a looping trace


• Complete or unbounded model checking explores all traces of the transition 
system


• Bounded model checking explores all traces up to a given maximum number of 
transitions before looping back



Verification
• check commands can be used to verify temporal assertions


• The default verification mechanism is bounded model checking


- The default maximum number of transitions is 10


- This can be changed by setting a scope for steps


• Alloy 6 also supports unbounded model checking


- Activated by the special scope 1.. steps

- Requires model checkers nuXmv or NuSMV to be installed



Future temporal operators

always           will always be true
eventually       will eventually be true
after            will be true in the next state
 until          will eventually be true and  is true until then

 releases       can only stop being true after 

ϕ ϕ
ϕ ϕ

ϕ ϕ
ψ ϕ ϕ ψ
ϕ ψ ψ ϕ



Future operators

ϕ ϕ ϕ ϕ ϕ ϕ ϕϕ

always ϕ

ϕ

eventually ϕ

ϕ

ϕ

Future operators

ψ ψ ϕψ

 until ψ ϕ

ϕ

ψ ψ  ψ
ϕψ

 releases ϕ ψ

ψ ψ ψ ψ ψ ψ ψψ



Mixing operators

 ϕ
ψ ψ ψ ψ ψ

always (  implies always )ϕ ψ

ϕ ψ ϕ  ϕ
ψ

always (  implies eventually )ϕ ψ

ϕ ψ ϕ ψ

always (  implies after )ϕ ψ



Mixing operators

ϕ ϕ ϕ ϕ

always (eventually )ϕ

ϕ ϕ ϕ ϕ ϕ

eventually (always )ϕ



Past temporal operators

historically     was always true
once             was once true
before           was true in previous state

 since          was once true and  was true since then

 triggered      was always true back to the point where  was true

ϕ ϕ
ϕ ϕ

ϕ ϕ
ψ ϕ ϕ ψ
ϕ ψ ψ ϕ



Past operators

ϕ ϕ ϕ  ϕ

always (  implies historically )ψ ϕ

ϕ ψ

always (  implies once )ψ ϕ

 ϕ
ψ

ϕ ψ ϕ ψ

always (  implies before )ψ ϕ



Past operators

ϕ θ θ  θ
ψ

always (  implies  since )ψ θ ϕ

 ϕ
ψ

 ϕ
θ θ θ  θ

ψ

always (  implies  triggered )ψ ϕ θ

θ θ  θ
ψθ



Safety properties

• Safety properties prevent some undesired behaviors from happening


- Easier to model check, since it suffices to search for a finite sequence 
of steps that leads to a bad state


- It is irrelevant what happens afterwards, and any continuation leads to a 
counter-example


- The archetypal safety property is an invariant specified as always ϕ



Liveness properties

• Liveness properties force some desired behaviors to happen


- Harder to model check, since it is necessary to search for a complete 
infinite trace where the desired behavior never happened


- Harder to specify, since they require fairness assumptions that prevent 
the system from stuttering forever


- The archetypal liveness property is eventually ϕ



Some operational principles
check invariant {

// No item can simultaneously be accessible and trashed

always no Accessible & Trashed

}

check restore_after_delete {

// A restore is only possible after a delete

all x : Item | always (restore[x] implies once delete[x])

}

check accessible_after_delete {

// A deleted item only becomes accessible again after being restored or created

all x : Item | always {

delete[x] implies after {

(restore[x] or create[x]) releases x not in Accessible

}

}

}



The key operational principles
pred can_restore [x : Item] { x in Trashed }

check delete_restore {

// After delete(x), can restore(x) and then x in accessible

all x : Item | always {

delete[x] implies after can_restore[x]

(delete[x]; restore[x]) implies x in Accessible''

}

} for 4 Item, 20 steps

pred can_empty { some Trashed }

check delete_empty {

// After delete(x), can empty() and then x not in accessible or trashed

all x : Item | always {

delete[x] implies after can_empty

delete[x] and after empty implies x not in (Trashed+Accessible)''

}

} for 4 Item, 20 steps



Verification

Label modeling a la Jackson
concept label [Item]
purpose
  organize items into overlapping categories
state
  labels : Item -> set Label
actions
  affix (i : Item, l : Label)
    when l not in the labels of i
    add l to the labels of i
  detach (i : Item, l : Label)
    when l in the labels of i
    remove l from the labels of l
  clear (i : Item)
    when i has some labels
    remove all labels of i
operational principle
  after affix(i,l), while no detach(i,l) and no clear(i), i is in the labels of l



The label in Alloy
sig Item {
  var labels : set Label
}
sig Label {}

fact Behavior {
  no labels
  always {
    (some i : Item, l : Label | affix[i,l] or detach[i,l])
    or 
    (some i : Item | clear[i])
    or
    stutter 
  }
}



Affix label with point-wise effect

pred affix [i : Item, l : Label] {
  // guard
  l not in i.labels
  // effect
  i.labels' = i.labels + l
  // frame condition
  all j : Item - i | j.labels' = j.labels
}



Affix label with point-free effect

pred affix [i : Item, l : Label] {
  // guard
  l not in i.labels
  // effect
  labels' = labels + i->l
}



Detach label

pred detach [i : Item, l : Label] {
  // guard
  l in i.labels
  // effect
  labels' = labels - i->l
}



Clear item

pred clear [i : Item] {
  // guard
  some i.labels
  // effect
  labels' = labels - i->Label
}



Label scenarios
run Scenario1 {
  some i : Item, disj l,m : Label {
    affix[i,l]; affix[i,m]; clear[i]
  }
} expect 1

run Scenario2 {
  some i : Item, l : Label {
    affix[i,l]; affix[i,l]
  }
} expect 0



Label operational principle

check affix_find {

// after affix(i,l), while no detach(i,l) and no clear(i), i is in the labels of l

all i : Item, l : Label | always {

    affix[i,l] implies after ((detach[i,l] or clear[i]) releases l in i.labels)

}

}



App design



Modularizing concepts

• To enable reuse and instantiation each concept should be in a 
parametrized module


• The module can still be used on its own, as Alloy implicitly declares 
parameter signatures


• Since a parameter signature cannot be extended with new fields, some 
tricks might be necessary to declare them



Trash

module Trash [Item]

sig Item {}
var sig Accessible in Item {}
var sig Trashed in Item {}

...



Label
module Label [Item]

sig Item {
  var labels : set Label
}
sig Aux in Item {

var labels : set Label
}
fact { Aux = Item }
sig Label {}

...



Specifying apps
• Import the required concepts, instantiating parameter signatures as 

needed


• Compose the concepts


- Enforce interleaving, by requiring at most one concept not to stutter


- Synchronize actions as needed


• Validate, validate, validate


• Check some expected properties



A filesystem app

• Composed of trash and label


• Many options to explore


- When to allow affixing labels?


- When to delete labels?


- Whether to use special labels?



Free composition
open Trash[File] as trash
open Label[File] as label

sig File {}

fact Interleave {
  always {
    trash/stutter or
    label/stutter
  }
}

run Example {}



Filesystem v1

• Allow labelling only when accessible


• Clear labels when file is deleted



Filesystem v1

fact Synchronization {

  // allow affixing only if file is accessible

  all f : File, l : Label | always (affix[f,l] implies f in Accessible)

  // clear all labels after file is deleted

  all f : File | always (delete[f] and some f.labels implies after clear[f])

}



Filesystem v1
run Scenario1 {
  some f : File, l : Label {
    create[f]; affix[f,l]; delete[f]
  }
} expect 1

run Scenario2 {
  some f : File, l : Label {
    create[f]; delete[f]; affix[f,l] 
  }
} expect 0



Filesystem v1

DEMO



Filesystem v2

• Allow labelling when accessible or trashed


• Clear labels when trash is emptied



Filesystem v2
fact Synchronization {

  // allow labelling when accessible or trashed

  all f : File, l : Label | always (affix[f,l] implies f in Accessible+Trashed)

  // clear labels when trash is emptied

  always {

    empty implies after {

  (some f : File-Accessible | clear[f]) until no (File-Accessible).labels

 }

  }

}



Filesystem v2
run Scenario1 {
  some f : File, l : Label {

create[f]; affix[f,l]; delete[f]
}

} expect 1

run Scenario2 {
  some f : File, l : Label {

create[f]; delete[f]; affix[f,l]
}

} expect 1



Filesystem v2
run Scenario3 {
  some f : File, l : Label {

create[f]; delete[f]; empty; affix[f,l]
}

} expect 0

run Scenario4 {
  some disj f1,f2 : File, l : Label {

create[f1]; create[f2]; delete[f1]; affix[f2,l]; delete[f2]; affix[f1,l]; empty
}

} expect 1

run Scenario5 {
  some disj f1,f2 : File, l : Label {

create[f1]; delete[f1]; affix[f1,l]; empty; create[f2]
}

} expect 0



Filesystem v2

DEMO



Filesystem v3

• Allow labelling when accessible or trashed


• Clear labels when trash is emptied


• Affix special label Trashed when file is deleted


• Detach special label Trashed when file is restored



Filesystem v3
one sig Dirty extends Label {}

fact Synchronization {

  // allow labelling when accessible or trashed

  all f : File, l : Label | always (affix[f,l] implies f in Accessible+Trashed)

  // clear labels when trash is emptied

  always {

    empty implies after ((some f : File-Accessible | clear[f]) until no (File-Accessible).labels)

  }

  // affix label Trashed after delete

  all f : File | always (delete[f] and Dirty not in f.labels implies after affix[f,Dirty])

  // detach label Trashed after restore

  all f : File | always (restore[f] and Dirty in f.labels implies after detach[f,Dirty])

}



Filesystem v3

DEMO



Epilogue



–Daniel Jackson

“Software is built on abstractions. Pick the right ones, and programming 
will flow naturally from design [...] Pick the wrong ones, and 

programming will be a series of nasty surprises”



–Leslie Lamport

“A specification is an abstraction. [...] But I don’t know how to teach 
you about abstraction. A good engineer knows how to abstract the 

essence of a system and suppress the unimportant details when 
specifying and designing it. The art of abstraction is learned only 

through experience.” 



–Tony Hoare

“There are two ways of constructing a software design: One way is to 
make it so simple that there are obviously no deficiencies, and the other 
way is to make it so complicated that there are no obvious deficiencies. 

The first method is far more difficult.”



–Alan Perlis

Epigram 31 
“Simplicity does not precede complexity, but follows it.” 



always eventually some Alloy


